AD-A251 337

NRL/FR/6851—92-9381

Computer Aided Approach to the Design of Y-Junction Stripline and Microstrip Ferrite Circulators

R. E. NEIDERT

Microwave Technology Branch
Electronics Science and Technology Division

May 5, 1992

DTIC
ELECTE
JUN 17, 1992
S B D

92-15816

92 6 17 017

Approved for public release; distribution unlimited.

REPORT DOCUMENTATION PAGE

Form Approved
OMB No 0704-0188

Public reporting burden for this collection of information is estimated to average 1 hour per response including the time for reviewing instructions searching existing data sources gathering and maintaining the data needed, and completing and reviewing the collection of information Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden to Washington Headquarters Services, Directorate for Information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington, VA 22202-4302 and to the Office of Management and Budget Paperwork Reduction Project (0704-0188), Washington DC 20503

1. AGENCY USE ONLY *(Leave blank)*

2. REPORT DATE
May 5, 1992

3. REPORT TYPE AND DATES COVERED

4. TITLE AND SUBTITLE
Computer Aided Approach to the Design of Y-Junction Stripline and Microstrip Ferrite Circulators

5. FUNDING NUMBERS
PE - 62234N
PR - RS34R2
WU - 2535-0

6. AUTHOR(S)
Robert E. Neidert

7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES)
Naval Research Laboratory
Washington, DC 20375-5000

8. PERFORMING ORGANIZATION REPORT NUMBER
NRL/FR/6851—92-9381

9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES)
Chief of Naval Research
Arlington, VA 22217-5000

10. SPONSORING/MONITORING AGENCY REPORT NUMBER

11. SUPPLEMENTARY NOTES

12a. DISTRIBUTION/AVAILABILITY STATEMENT
Approved for public release; distribution is unlimited.

12b. DISTRIBUTION CODE

13. ABSTRACT *(Maximum 200 words)*

New interest is shown in modernization of ferrite component technology. The workhorse of ferrite components is the Y-junction circulator; it is used as a 3-port circulator, 2-port isolator, or a single-pole double-throw switch. For new technology it is useful to increase bandwidth capability, minimize size, and create designs compatible with monolithic circuits. Until now there has been no publicly available circulator analysis computer program.

A concise compendium on microwave ferrite Y-junction circulators is provided in this report. It consists of background and tutorial information on ferrite media, junction circulator theory and evolution, modern improvements in the theory that permit wideband analysis, inclusion of dielectric losses in the ferrite, and a listing for a FORTRAN computer program to perform wideband circulator performance analysis.

This theoretical information and computer program will be useful for further work on optimization of circulator bandwidth, size, and monolithic circuit compatibility. Also, it will be useful in examining quantitatively the effects of dielectric losses.

14. SUBJECT TERMS
Circulator microwave ferrite
Computer program FORTRAN model analysis
Microstrip stripline

15. NUMBER OF PAGES
24

16. PRICE CODE

17. SECURITY CLASSIFICATION OF REPORT	18. SECURITY CLASSIFICATION OF THIS PAGE	19. SECURITY CLASSIFICATION OF ABSTRACT	20. LIMITATION OF ABSTRACT
UNCLASSIFIED	UNCLASSIFIED	UNCLASSIFIED	SAR

NSN 7540-01-280-5500
Standard Form 298 (Rev 2-89)
Prescribed by ANSI Std Z39-18
298-102

CONTENTS

Accession For
NTIS GRA&I ☑
DTIC TAB ☐
Unannounced ☐
Justification

By
Distribution/
Availability Codes
Dist | Avail and/or Special
A-1

COMPUTER AIDED APPROACH TO THE DESIGN OF Y-JUNCTION STRIPLINE AND MICROSTRIP FERRITE CIRCULATORS

1. INTRODUCTION

New interest is evident in modernization of ferrite component technology. Monolithic microwave circuit technology, which was perfected in the 1980s, has permitted the development of multidecade bandwidth amplifiers, multioctave bandwidth subsystem components, and full systems that have more than an octave of bandwidth. The Y-junction circulator is the workhorse of ferrite components. It can be used as a 3-port circulator, 2-port isolator, or a single-pole, double-throw switch. For new circulator technology it would be useful to increase the bandwidth capability (currently about an octave), minimize the size (perhaps by using matching circuits other than quarter-wavelength transformers), and create design approaches that are compatible with monolithic circuits. The first step is to have on hand an accurate theoretical analysis tool to calculate intrinsic circulator performance, thereby permitting simultaneous circulator and matching circuit optimization. The following is an effort to assemble a concise circulator compendium that consists of background and tutorial information on ferrite media, wave propagation in ferrite media, junction circulator theory and evolution, modern improvements in the theory to permit wideband analysis, and finally a computer program to perform the wideband analysis.

The landmark theoretical works on Y-junction circulators are those of:

- Bosma [1] (solves the simplified boundary-value problem of the circulator ferrite disk by using a Green's function approach that relates the axial component of the electric field to the circumferential component of the magnetic field at the perimeter of the disk);

- Fay and Comstock [2] (describe the operation in terms of counter-rotating propagation modes in the ferrite disk);

- Wu and Rosenbaum [3] (predict octave bandwidth microstrip circulator operation theoretically using Bosma's Green's function analysis); and

- Schloemann and Blight [4] (rederive the Bosma form of the Green's function for $\mu_{eff} < 0$). When the bias field is low,

$$\mu_{eff} = (\mu^2 - \kappa^2) / \mu \tag{1}$$

is real, where μ and κ are the off-diagonal components of the ferrite material's permeability tensor. Mu-effective (μ_{eff}) becomes real and negative at frequencies well below the circulator design center frequency, at and below the point where

$$f = (\gamma / 2\pi) \cdot 4\pi M_S, \tag{2}$$

Manuscript approved March 8, 1992

in which f is the frequency, γ is the gyromagnetic ratio of electron magnetic moment to angular momentum, $(\gamma/2\pi)$ is approximately 2.8 MHz/Oersted, and $4\pi M_S$ is the ferrite material's saturation magnetization.

The new theoretical work reported here includes the derivation of the analytical link between the results of Refs. [3,4], the inclusion of the imaginary part of the dielectric permittivity of the ferrite material in the calculations, and the development of the computer program for quantitative analyses. The first item facilitates the use of a set of published equations for part of the computer programming; and the second permits the effects of dielectric losses in the ferrite material to be calculated.

Some of the most important experimental stripline-junction circulator work has been that of Simon [5], which gives the results of tests on a wide range of materials; Salay and Peppiatt [6], which clarifies the sign of the reactive part of the input impedance; and Schloemann and Blight [4], which shows very wideband performance over 3:1 bands.

2. THEORY

(a) Ferrite Medium

A biased ferrite material has complex permittivity $\boldsymbol{\varepsilon}$ and tensor permeability $\overset{\leftrightarrow}{\boldsymbol{\mu}}$, and its elements may be complex. The tensor character of the permeability is developed from the Landau-Lifshitz equation of motion of the magnetization vector **M** in the material

$$\frac{\partial \mathbf{M}}{\partial t} = \gamma\,(\mathbf{M}\times\mathbf{H}) + \left[\frac{\gamma\,\alpha\,(\mathbf{M}\times\mathbf{H})\times\mathbf{M}}{|\mathbf{M}|}\right], \tag{3}$$

with a small RF signal applied. **M** is magnetization, **H** is magnetic field, and α is an experimentally determined parameter that adjusts the magnitude of the bracketed damping term. In rectangular coordinates, with the magnetic bias field in the z-direction, **H** and **M** are

$$\mathbf{H} = \mathbf{i}\,h_x + \mathbf{j}\,h_y + \mathbf{k}\,H_B \tag{4}$$

and

$$\mathbf{M} = \mathbf{i}\,m_x + \mathbf{j}\,m_y + \mathbf{k}\,M_B\,, \tag{5}$$

where $m_x << M_B$ and $m_y << M_B$, with the lower case letters indicating the RF values and the capital letters indicating the bias values. (Usually H_B is large enough to make M_B equal to the saturation value M_s.) The solution has the form

$$m_{x_0} = \left[\frac{\gamma^2 M_B H_B + \lambda\left(j\omega + \lambda H_B / M_B\right)}{\gamma^2 H_B^2 + \left(j\omega + \lambda H_B / M_B\right)^2}\right] h_{x0} - \left[\frac{\gamma\left[M_B\left(j\omega + \lambda H_B / M_B\right) - \lambda H_B\right]}{\gamma^2 H_B^2 + \left(j\omega + \lambda H_B / M_B\right)^2}\right] h_{y0}, \tag{6}$$

where $\lambda = -\gamma\alpha\ |\mathbf{M}|$ and $|\mathbf{M}| \approx M_B$, with the zero subscripts indicating that the harmonic time dependence is not included in the values ($m_x = m_{x0}e^{j\omega t}$). By convention, Eq. (6) is written:

$$m_{x_0} = \chi_{xx}\,h_{x0} - j\chi_{yx}\,h_{y0}\,, \tag{7}$$

and the similar solution for m_{y_0} is :

$$m_{y0} = j\chi_{xy}\, h_{x0} + \chi_{yy}\, h_{y0}\,. \tag{8}$$

In the solutions $\chi_{yy} = \chi_{xx}$ and $\chi_{yx} = -\chi_{xy}$ so that the intrinsic RF magnetic susceptibility tensor can be written as

$$\overleftrightarrow{\chi} = \begin{bmatrix} \chi_{xx} & -j\chi_{xy} & 0 \\ j\chi_{xy} & \chi_{xx} & 0 \\ 0 & 0 & 0 \end{bmatrix}, \tag{9}$$

assuming m_z is negligibly small relative to M_B. In the CGS electromagnetic system of units used in this part of the derivation,

$$\mathbf{b} = \mathbf{h} + 4\,\pi\,\mathbf{m} = (\,1 + 4\,\pi\,\overleftrightarrow{\chi})\,\mathbf{h} = \overleftrightarrow{\mu}\,\mathbf{h}\,, \tag{10}$$

where

$$\mathbf{h} = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} h_x \\ h_y \\ h_z \end{bmatrix}. \tag{11}$$

So

$$\overleftrightarrow{\mu} = \begin{bmatrix} 1 + 4\pi\,\chi_{xx} & -j4\pi\,\chi_{xy} & 0 \\ j4\pi\,\chi_{xy} & 1 + 4\pi\,\chi_{xx} & 0 \\ 0 & 0 & 1 \end{bmatrix}, \tag{12}$$

which may be written as

$$\overleftrightarrow{\mu} = \begin{bmatrix} \mu & -j\kappa & 0 \\ j\kappa & \mu & 0 \\ 0 & 0 & 1 \end{bmatrix}. \tag{13}$$

This is the intrinsic permeability tensor of the ferrite material and is related to material parameters, bias field, and frequency as shown in Eq. (6),

where

$$\mu = 1 + 4\,\pi\,\chi_{xx} = \mu' - j\,\mu'', \tag{14}$$

and

$$\kappa = 4\,\pi\,\chi_{xy} = \kappa' - j\,\kappa''. \tag{15}$$

(b) Plane Waves in Magnetized Ferrite Media

For an RF field in any linear, isotropic, homogeneous medium

$$\left.\begin{aligned} \nabla\times\mathbf{E} &= -\frac{\partial\mathbf{b}}{\partial t} \\ \nabla\times\mathbf{h} &= -\frac{\partial\mathbf{D}}{\partial t} \end{aligned}\right\} \tag{16}$$

with $\mathbf{D} = \varepsilon\,\mathbf{E}$ and $\mathbf{b} = \overleftrightarrow{\mu}\,\mathbf{h}$. Consider plane waves of the form

$$(\mathbf{E}\ ,\ \mathbf{h}\ ,\ \mathbf{b}\ ,\ \mathbf{D}\ ) = (\ \mathbf{E_0}\ ,\ \mathbf{h_0}\ ,\ \mathbf{b_0}\ ,\ \mathbf{D_0}\ )\ e^{j\,\omega t - \Gamma \mathbf{n} \bullet \mathbf{r}}\ , \tag{17}$$

where the 0 subscripted values do not have time and coordinate dependence, where **n** is a unit vector in the direction of plane wave propagation, and **r** is related to the system coordinates ($\mathbf{r} = \mathbf{i}x + \mathbf{j}y + \mathbf{k}z$ in rectangular coordinates). Solving Eq. (16) in magnetic terms that involve $\overleftrightarrow{\boldsymbol{\mu}}$ yields

$$\omega^2\ \varepsilon\ \mathbf{b}_0 = \Gamma^2\ \mathbf{n} \times (\ \mathbf{n} \times \mathbf{h}_0\ ) \tag{18}$$

which may be solved for Γ, the propagation constant.

The propagation constant is

$$\Gamma_\pm = j\omega\sqrt{\varepsilon\mu_0}$$

$$\left\{\frac{(\mu^2-\mu-\kappa^2)\sin^2\theta + 2\,\mu \pm \left[(\mu^2-\mu-\kappa^2)\sin^4\theta + 4\,\kappa^2\cos^2\theta\right]^{1/2}}{2\left[(\mu-1)\sin^2\theta + 1\right]}\right\}^{1/2} . \tag{19}$$

The term inside the braces is the relative effective permeability that depends on θ, the angle between the direction of propagation and the magnetic bias field, as well as on the material parameters contained in $\varepsilon = \varepsilon_0\ \varepsilon_r$, μ_0 , μ , and κ, and on frequency contained in $\omega = 2\pi f$. For propagation perpendicular to H_B, as is the case in a junction circulator "puck" of ferrite material,

$$\Gamma_\pm = j\omega\sqrt{\varepsilon\mu_0}\left\{\frac{(\mu^2-\mu-\kappa^2) + 2\,\mu \pm (\mu^2-\mu-\kappa^2)}{2\,\mu}\right\}^{1/2} = j\omega\sqrt{\varepsilon\mu_0}\,\{\mu_{eff}\}^{1/2} . \tag{20}$$

Then

$$\mu_{eff+} = (\ \mu^{\,2} - \kappa^{\,2})\ /\ \mu\ , \tag{21}$$

which corresponds to a linearly polarized wave with the **h** field perpendicular to H_B. Also,

$$\mu_{eff-} = 1, \tag{22}$$

which corresponds to a linearly polarized wave with the **h** field parallel to H_B. It propagates as if in air except for permittivity effects. In the junction circulator, μ_{eff+} is the operative term.

The preceding theoretical discussion is sufficient background to proceed to the special problem of Y-junction circulators. General information on additional theoretical matters and on other kinds of ferrite applications is provided by Soohoo [7], and by Linkhart [8].

(c) Stripline and Microstrip Junction Circulators

The field problem for the 3-port symmetrical Y-junction stripline circulator has been addressed by Bosma [1] by using the notation shown in Fig. 1. The basic assumptions in the analysis were:



Fig. 1 – Circulator coordinate reference

(1) Stripline configuration: Two equal thickness ferrite disks between two ground planes and a zero thickness center conductor disk of the same radius with three equally spaced coupling lines of width w, as shown in Fig. 1. Bosma notes that, as is the case in striplines, the electromagnetic fields are equal in magnitude and opposite in polarity on opposite sides of the center conductor, so that the field problem in the ferrite need be solved for only one disk. Therefore, his solution is approximately valid for microstrip.

(2) The fields in the disk do not depend on the z coordinate, so that the solution need be only 2-dimensional (2-D).

(3) Except at the connections of the coupling striplines, no radial current can flow from the edge of the center conductor.

(4) The tangential component of $\mathbf{H}_\phi\ (R\ ,\ \phi)$ in the ferrite is equal to the magnetic field intensity in the corresponding coupling stripline at the same place.

(5) $\mathbf{H}_\phi\ (R\ ,\ \phi)$ is constant over the coupling stripline width w, and it is 0 elsewhere. In the cylindrical coordinate system, the values of $\mathbf{H}_\phi\ (R\ ,\ \phi)$ are H_a at the input port, H_b at the output port, H_c at the isolated port, and 0 elsewhere. In general H_a, H_b, and H_c are complex.

(6) The electric field intensity in the disk has a z component only, $E_z\ (r, \phi)$, or $E_z(R, \phi)$ at the disk perimeter.

In Ref. [1] Bosma indicates that $E_z\ (r, \phi)$ satisfies the Helmholtz vector wave equation

$$\nabla^2\mathbf{E} = \mu\varepsilon \frac{\partial^2 \mathbf{E}}{\partial t^2} \tag{23}$$

or assuming harmonic time dependence

$$\nabla^2 \mathbf{E} = -\omega^2 \mu\varepsilon \, \mathbf{E} \, . \tag{24}$$

So, in cylindrical coordinates

$$\left[\frac{\partial^2 E_z}{\partial r^2} + \frac{1}{r}\frac{\partial E_z}{\partial r} + \frac{1}{r^2}\frac{\partial^2 E_z}{\partial \phi^2} + (\omega \sqrt{\mu\varepsilon})^2 E_z \right] = 0 \, . \tag{25}$$

From Maxwell's equation assuming harmonic time dependence,

$$\nabla \times \mathbf{E} = j\omega\mu_0 \overleftrightarrow{\mu} \, \mathbf{H} \tag{26}$$

comes

$$\begin{bmatrix} \frac{1}{r}\frac{\partial E_z}{\partial \phi} \\ -\frac{\partial E_z}{\partial r} \\ 0 \end{bmatrix} = j\omega\mu_0 \begin{bmatrix} \mu & -j\kappa & 0 \\ j\kappa & \mu & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} H_r \\ H_\phi \\ H_z \end{bmatrix} \tag{27}$$

giving

$$H_r = \frac{-j\frac{1}{r}\frac{\partial E_z}{\partial \phi} - \frac{\kappa}{\mu}\frac{\partial E_z}{\partial r}}{\omega\mu_0\mu_{eff}} \tag{28}$$

and

$$H_\phi = \frac{j\frac{1}{r}\frac{\partial E_r}{\partial r} - \frac{\kappa}{\mu}\frac{1}{r}\frac{\partial E_r}{\partial \phi}}{\omega\mu_0\mu_{eff}} \, . \tag{29}$$

A Green's function is then introduced that relates H_ϕ and E_z at the edge of the disk:

$$E_z(R,\phi) = \int_{-\pi}^{\pi} G(R,\phi \, ; R,\phi') \, H_\phi(R,\phi') \, d\phi'. \tag{30}$$

For ϕ in radians,

$$H_\phi(R,\phi) = \begin{Bmatrix} H_a & \text{at}\left(-\frac{\pi}{3}-\Psi\right) \langle\ \phi\ \langle \left(-\frac{\pi}{3}+\Psi\right) \\ H_b & \text{at}\left(\frac{\pi}{3}-\Psi\right) \quad \langle\ \phi\ \langle \quad \left(\frac{\pi}{3}+\Psi\right) \\ H_c & \text{at}\left(\pi-\Psi\right) \quad \langle\ \phi\ \langle \quad \left(\pi+\Psi\right) \\ \text{Zero} & \text{everywhere else} \end{Bmatrix}. \tag{31}$$

Then, using $G\ (\phi\ ;\ \phi')$ for $G(R,\phi\ ;\ R,\phi')$, and for small ψ,

$$\left.\begin{aligned} E_z\!\left(R\ ,-\tfrac{\pi}{3}\right) &= E_A = 2\,\Psi\left[G\!\left(-\tfrac{\pi}{3};-\tfrac{\pi}{3}\right)\!H_a + G\!\left(-\tfrac{\pi}{3};\tfrac{\pi}{3}\right)\!H_b + G\!\left(-\tfrac{\pi}{3};\pi\right)\!H_c\right] \\ E_z\!\left(R\ ,\tfrac{\pi}{3}\right) &= E_B = 2\,\Psi\left[G\!\left(\tfrac{\pi}{3};-\tfrac{\pi}{3}\right)\!H_a + G\!\left(\tfrac{\pi}{3};\tfrac{\pi}{3}\right)\!H_b + G\!\left(\tfrac{\pi}{3};\pi\right)\!H_c\right] \\ &\text{and} \\ E_z\!\left(R\ ,\pi\right) &= E_C = 2\,\Psi\left[G\!\left(\pi\,;-\tfrac{\pi}{3}\right)\!H_a + G\!\left(\pi\,;\tfrac{\pi}{3}\right)\!H_b + G(\pi\,;\pi)H_c\right] \end{aligned}\right\}. \tag{32}$$

Note that E_A, E_B, and E_C are the electric field intensities at each of the three circulator ports: input, output, and isolated, respectively.

Bosma [1] notes that a solution to Eq. (25) is a series with

$$E_{z,n}(r,\phi) = a_n\, J_n(kr)\, e^{jn\phi}\ , \tag{33}$$

where J_n is the n^{th} order Bessel function and k is the propagation constant $\omega(\mu_0\, \varepsilon_0\, \mu_{eff}\, \varepsilon_f)^{1/2}$, where μ_{eff} and ε_f refer to the ferrite material. Then from Eq. (29),

$$H_{\phi,n}(r,\phi) = j\,\frac{a_n}{Z_{eff}}\left[J_n'(kr) - \frac{\kappa}{\mu}\frac{n\,J_n(kr)}{kr}\right] e^{jn\phi}\ , \tag{34}$$

where $Z_{eff} = (\mu_0\, \mu_{eff}\, /\, \varepsilon_0\, \varepsilon_f)^{1/2}$ is a wave impedance in the ferrite. Then, at the disk edge, if

$$H_{\phi,n}(R,\phi) = b_n\, e^{jn\phi} \tag{35}$$

then

$$E_{z,n}(r,\phi) = j\, b_n\, \frac{Z_{eff}\, J_n(kr)\, e^{jn\phi}}{\frac{\kappa}{\mu}\frac{n\,J_n(kr)}{kr} - J_n'(kr)}. \tag{36}$$

Finally, the Green's function is

$$
\left.\begin{aligned}
G(r,\phi\,;R,\phi') &= X + Y \\
X &= \frac{-j\,Z_{eff}\,J_0(kr)}{2\,\pi\,J_0'(kr)} \\
Y &= \frac{Z_{eff}}{\pi}\sum_{n=1}^{\infty}\frac{\frac{\kappa}{\mu}\frac{n\,J_n(kR)}{kR}\sin n(\phi-\phi') - j\,J_n'(kR)\cos n(\phi-\phi')}{\left[J_n'(kR)\right]^2-\left[\frac{\kappa}{\mu}\frac{n\,J_n(kR)}{kR}\right]^2}
\end{aligned}\right\} \tag{37}
$$

(d) Computer Program

Using Eqs. (31) and (32), Wu and Rosenbaum have solved for H_ϕ and E_z to obtain an input wave impedance E_z/H_ϕ, and the full scattering matrix for the 3-port circulator:

$$
Z_{in_{wave}} = -Z_d - \left(\frac{j\,2\,Z_{eff}}{\pi}\right)\left(\frac{D}{C_1^2 - C_2\,C_3}\right), \tag{38}
$$

$$
\left.\begin{aligned}
S_{11} = S_{22} = S_{33} &= 1 + \frac{\pi\,Z_d\left(C_1^2 - C_2\,C_3\right)}{j\,Z_{eff}(D)} \\
S_{21} = S_{13} = S_{32} &= \frac{\pi\,Z_d\left(C_2^2 - C_1\,C_3\right)}{j\,Z_{eff}(D)} \\
S_{31} = S_{23} = S_{12} &= \frac{\pi\,Z_d\left(C_3^2 - C_1\,C_2\right)}{j\,Z_{eff}(D)}
\end{aligned}\right\}. \tag{39}
$$

Here Z_d is a normalizing impedance (of a plane wave in the medium just outside the ferrite disk),

$$
Z_d = 377\ \Omega\,/\,(\varepsilon_d)^{1/2}, \tag{40}
$$

where ε_d is the relative dielectric constant of the region just outside the disk, and

$$
D = C_1^3 + C_2^3 + C_3^3 - 3\,C_1\,C_2\,C_3 \tag{41}
$$

The values of $C1$, $C2$, and $C3$ are

$$
\left.\begin{aligned}
C_1 &= \frac{\Psi\,B_0}{2\,A_0} + \sum_{n=1}^{\infty}\left(\frac{\sin^2 n\Psi}{n^2\Psi}\right)\frac{A_n\,B_n}{D_n} - \frac{\pi\,Z_d}{j\,2\,Z_{eff}} \\
C_2 &= \frac{\Psi\,B_0}{2\,A_0} + \sum_{n=1}^{\infty}\left(\frac{\sin^2 n\Psi}{n^2\Psi}\right)\frac{A_n\,B_n\cos\left(\frac{2\,n\,\pi}{3}\right) - \left(j\,\frac{n\kappa}{\mu kR}\right)B_n^2\sin\left(\frac{2\,n\,\pi}{3}\right)}{D_n} \\
C_3 &= \frac{\Psi\,B_0}{2\,A_0} + \sum_{n=1}^{\infty}\left(\frac{\sin^2 n\Psi}{n^2\Psi}\right)\frac{A_n\,B_n\cos\left(\frac{2\,n\,\pi}{3}\right) + \left(j\,\frac{n\kappa}{\mu kR}\right)B_n^2\sin\left(\frac{2\,n\,\pi}{3}\right)}{D_n}
\end{aligned}\right\}. \tag{42}
$$

where

$$D_n = A_n^2 - \left(\frac{n\kappa}{\mu kr}\right)^2 B_n^2 \text{ , with } A_n = J_n'(kR) \text{ and } B_n = J_n(kR) \text{ .}$$

This infinite series solution is not actually correct since it assumes an abrupt step of the *RF* fields at the edges of the three input/output transmission lines. The abrupt field change cannot actually happen; therefore, some number of terms less than infinity is more nearly accurate. Bosma [1] speaks of a simulation using 18 terms, Wu and Rosenbaum [3] use three terms, Schloemann and Blight [4] use nine terms. In many test calculations using the program developed here it has been observed that less than 5% difference exists in the performance parameters between using three terms and using nine terms. The Wu and Rosenbaum [3] equations can be programmed almost directly. The input impedance, however, should be denormalized from Z_d and instead normalized to the characteristic impedance of the input/output stripline or microstrip lines. Also, the sign of the input reactance must be reversed as indicated by Salay and Peppiatt [6].

(e) Computer Program Allowing Negative Effective Permeability

Most circulators above about 2 GHz are biased at an external magnetic field in the vicinity of $4\pi Ms$, which is a field level below the one required to produce the ferromagnetic loss peak at or near the circulator center frequency; this is referred to as *below resonance* operation. The larger field necessary to operate *above resonance* is generally not physically practical above 2 GHz. The internal field values are small in a ferrite circulator disk that is biased below resonance because of the demagnetizing effect of the geometrical shape of the disk. The demagnetizing factor with the bias magnetic field perpendicular to a disk is $4\pi Ms$, so that

$$H_{\text{int}} = H_{\text{ext}} - 4\pi Ms \text{ .} \tag{43}$$

With relatively small internal bias field, and with operating frequency far above the ferromagnetic resonance frequency, the imaginary parts of μ and κ approach zero, the real part of μ approaches unity, and the real part of κ is approximately $[-(\gamma/2\ \pi) \bullet 4\ \pi\ M_s/f]$, as shown in Fig. 5 of Ref. [2].

Schloemann and Blight [4] note that when the magnitude of κ becomes larger than unity, which happens at about $f_m = (\gamma/2\pi) \bullet 4\pi M_S$, then $\mu_{eff} = (\mu^2 - \kappa^2)/\mu$ becomes negative. To make calculations possible below f_m they rederived the Green's function terms for $\mu_{eff} < 0$. The result is that V_o, V_n, and U_n of the new Green's function, written as

$$G(\phi;\phi') = V_0 + \sum_{n=1}^{\infty} V_n \cos n(\phi - \phi') + U_n \sin n(\phi - \phi'), \tag{44}$$

change such that each sign is reversed, each Bessel function $J_n(kR)$ becomes a modified Bessel function $I_n(kR)$ of the same kind and order, each Bessel function derivative becomes a modified Bessel function derivative of the same kind and order, and the positive of μ_{eff} is used.

To keep the straightforward computer program approach used by Wu and Rosenbaum to solve the circulator problem, it is necessary to find relations between the Green's function terms given in Refs. [1] and [4] and the $C1$, $C2$, and $C3$ terms of Ref. [3]. Writing Eq. (37) with $\phi = -\pi/3$ and $\phi' = -\pi/3$ corresponding to Port 1 of the circulator, with $r = R$,

$$G\left(-\frac{\pi}{3};-\frac{\pi}{3}\right)=\frac{-j\,Z_{eff}B_0}{2\,\pi A_0}+\frac{Z_{eff}}{\pi}\sum_{n=1}^{\infty}\frac{-j\,A_n\,B_n}{D_n}\,. \tag{45}$$

But

$$C_1=\frac{\Psi\,B_0}{2\,A_0}+\sum_{n=1}^{\infty}\left(\frac{\sin^2 n\Psi}{n^2\Psi}\right)\frac{A_n\,B_n}{D_n}-\frac{\pi\,Z_d}{j\,2\,Z_{eff}}. \tag{46}$$

Therefore

$$C_1=j\,\frac{\Psi\,\pi}{Z_{eff}}\left[G\left(-\frac{\pi}{3};-\frac{\pi}{3}\right)\right]+\frac{j\,\pi\,Z_d}{2\,Z_{eff}}\,. \tag{47}$$

This expression shows how $C1$ is modified when the terms of $G\ (-\pi/3\ ;-\pi/3)$ are modified for $f<f_m$. Similar correlations exist between $C2$ and $G\ (-\pi/3\ ;+\pi/3)$, and between $C3$ and $G\ (-\pi/3\ ;\pi)$, as follows:

$$C_2=j\,\frac{\Psi\,\pi}{Z_{eff}}\left[G\left(-\frac{\pi}{3};\frac{\pi}{3}\right)\right], \tag{48}$$

and

$$C_3=j\,\frac{\Psi\,\pi}{Z_{eff}}\left[G\left(-\frac{\pi}{3};\pi\right)\right]. \tag{49}$$

CIRCREN uses these modified expressions in its calculations.

There is a small difference between Refs. [3] and [4] in which Ref. [3] uses $(\sin^2 n\,\Psi/n^2\,\Psi)$in its summation of terms where Ref. [4] uses Ψ. In the limit the summation term approaches Ψ so that the two are interchangeable if n is reasonably large. Differences between the two were very minor in calculations that use n from 3 to 18. CIRCREN is set up to use the $(\sin^2 n\Psi/n^2\Psi)$ summation.

3. COMPUTER PROGRAM OPERATION

The theory described has been programmed in FORTRAN. The program is named CIRCREN. Appendix A lists the February 1992 version of the program and a sample run. It is currently set up for *below resonance* calculations, for applications above 2 GHz assuming that the bias field is approximately $4\pi Ms$. It calculates the characteristic impedance of stripline or microstrip transmission lines in and out. The number of Green's function terms is left to the user to supply, but 3 to 18 is a reasonable range. The user supplies the $4\pi Ms$, relative dielectric constant, and radius of the circulator puck, and the dielectric constant of the material outside the ferrite disk, along with the overall height and the width of the three input/output transmission lines. The subfunction BF calculates the Bessel functions directly from a series, switching to modified Bessel function calculation when necessary by means of the KEY value. Recursion relations are used to calculate the Bessel function derivatives.

Virtually exact agreement (with the sign of the input reactance reversed) has been ach'eved with Wu and Rosenbaum's calculations by using $n = 3$, and with Schloemann and Blight's calculations with negative μ_{eff} and with $n = 9$, as shown in Figs. 2 and 3, respectively. Good agreement with Salay and Peppiatt's calculations, compared with their measured results, is shown in Fig. 4; they conclude that an effective radius somewhat less than the actual radius would produce better calculated agreement with the measured results. These measured data, which agree with their own independent derivation, are also their corroboration that the sign of Bosma's reactive term is reversed from the correct sign. The calculations of CIRCREN use the sign of the reactance that agrees with Salay and Peppiatt's calculations and measurements. The low-frequency extension is believed to be valid only down to about $f_m/2$ as discussed in Ref. [4].



Fig. 2 – Comparison with Wu and Rosenbaum



Fig. 3 – Comparison with Schloemann and Blight



Fig. 4 – Comparison with Salay and Peppiatt

4. SUMMARY

The main objective here has been to produce a computer program to calculate the intrinsic performance of a Y-junction circulator from physical dimensions and magnetic properties of the ferrite material and embedding structure. This has been done, with the program set up for the most common case of *below resonance* operation. The mathematical manipulation described in Section 2(e) permitted the programming of the convenient equation forms described in Section 2(d).

Excellent agreement with other authors' calculations and with measured data in a carefully controlled experiment [6] has been shown. As a result, we can confidently say that this program can be used for further work on optimization of circulator bandwidth, size, and monolithic circuit compatibility.

REFERENCES

1. H. Bosma, "On Stripline Y-Circulation at UHF," *IEEE Trans. MT&T*, 61-72 (1964).

2. C.E. Fay and R.L. Comstock, "Operation of the Ferrite Junction Circulator," *IEEE Trans. MT&T*, 15-27 (1965).

3. Y.S. Wu and F.J. Rosenbaum, "Wide-Band Operation of Microstrip Circulators," *IEEE Trans. MT&T*, 849-856 (1974).

4. E. Schloemann and R.E. Blight, "Wide-Band Operation of Microstrip Circulators Based on YIG and Li-ferrite Single Crystals, " *IEEE Trans. MT&T*, 1394-1400 (1986).

5. J.W. Simon, "Broadband Strip-Transmission Line Y-Junction Circulators," *IEEE Trans. MT&T*, 335-345 (1965).

6. S.J. Salay and H.J. Peppiatt, "Input Impedance Behavior of Stripline Circulator," *IEEE Trans. MT&T*, 109-110 (1971).

7. R.F. Soohoo, *Theory and Application of Ferrites*. (Prentice-Hall, Inc., Englewood Cliffs, New Jersey, (1960).

8. D.K. Linkhart, *Microwave Circulator Design* (Artech House, Inc., Norwood, Massachusetts (1989)).

APPENDIX A

CIRCREN FORTRAN LISTING

```
C         CIRCREN CALCULATES THE PERFORMANCE OF A FERRITE CIRCULATOR,
C                      INCLUDING DIELECTRIC LOSS,
C                WITH MU-EFFECTIVE ALLOWED TO BE NEGATIVE.

      COMPLEX C11,C1,C2,C3,C4,C5,C6,C7
      COMPLEX SM,AL1,AL2,AL,BE,GA,ZIN,ZINWU,YIN
      COMPLEX EFC,ZEFF,SIC,ZIBCWU,ZIBC,SR,B0,A0
      COMPLEX Z,X9,Y9,RJ9,T9,S9,BF,C1A,C1C
      COMPLEX BN,AN,CK2,CK22,C1BD,C1B,C2BR,C2BI,AL11

      C=2.9979E10
      GAM=2.8E6
      PI=3.141593
      U0=1.2566E-8
      E0=8.8543E-14

50    WRITE(*,*)'FOR STRIPLINE ENTER 1 , FOR MICROSTRIP ENTER 2 :'
      READ(*,*)kstmi

            IF(KSTMI.EQ.1) GO TO 26
            IF(KSTMI.EQ.2) GO TO 26
            WRITE(*,*)'SL/MS CHOICE NOT 1 OR 2'
            GO TO 50

26    WRITE(*,*)'ENTER NUMBER OF GREENS FUNCTION TERMS TO USE :'
      READ(*,*)NGFT

      WRITE(*,*)'ENTER 4PIMS(G),DIEL CON FERR,LOSS TAN FERR :'
      READ(*,*)FPMS,EF,TAND

      WRITE(*,*)'ENTER DIEL CON OUTSIDE FERR :'
      READ(*,*)ED

      FM=GAM*FPMS
      ZD=120.*PI/SQRT(ED)

      WRITE(*,*)'ENTER DIMS RADIUS,LINE W,DIEL TH,LINE TH (ALL CM):'
      READ(*,*)R,W,B,T

      IF(KSTMI.EQ.1)THEN

C     CALCULATE STRIPLINE Z0

         IF(T.EQ.0.)T=B/1000000.
         X=T/B
         AM=2./(1.+2.*X/(3.*(1.-X)))
         DWBTA=X/PI/(1.-X)
         DWBTB=(X/(2.-X))**2+(.0796*X/((W/B)+1.1*X))**AM
         DWBT=DWBTA*(1.-.5*LOG(DWBTB))
         WPBT=(W/(B-T))+DWBT
         FSLA=(8./PI/WPBT)+SQRT((8./PI/WPBT)**2+6.27)

C     WU'S STRIPLINE GEOMETRIC FACTOR,F,FOLLOWS

         FSL=(.5/PI)*LOG(1.+(4./PI/WPBT)*FSLA)
         Z0SL=(ZD/2.)*FSL
         Z0=Z0SL

      ELSE

C     CALCULATE MICROSTRIP Z0

         ETA0=120.*PI
```

```
      IF(T.EQ.0.)T=B/1.E6

         IF((W/B).LE.(1./2./PI))THEN
           WPB=(W/B)+(1.25/PI)*(T/B)*(1.+LOG(4.*PI*W/T))
         ELSE
           WPB=(W/B)+(1.25/PI)*(T/B)*(1.+LOG(2.*B/T))
         ENDIF

         IF((W/B).LE.1.)THEN
            FWHL1=(1.+12.*B/W)**(-.5)+.04*(1.-W/B)**2.
            EE=((ED+1.)/2.)+((ED-1.)/2.)*FWHL1-(ED-1.)*T/B/4.6/SQRT(W/B)
            Z0MS=(ETA0/(2.*PI*SQRT(EE)))*LOG((8./WPB)+
+           .25*WPB)
         ELSE
            FWHG1=(1.+12.*B/W)**(-.5)
            EE=((ED+1.)/2.)+((ED-1.)/2.)*FWHG1-(ED-1.)*T/B/4.6/SQRT(W/B)
            Z0MS=(ETA0/SQRT(EE))/(WPB+1.393+.667*LOG(WPB+
+           1.444))
         ENDIF

      Z0=Z0MS

   ENDIF

   WRITE(*,*)'Z0= ',Z0

   SI=ASIN(W/2./R)

   WRITE(*,*)'ENTER (IN GHZ): FLOW,FHIGH,FSTEP'
   READ(*,*)FREQL,FREQH,FREQS

     WRITE(*,*)'   FREQ        RIN       XIN      RTN LS    ISOL      INS
+LS      GIN       BIN'


         F=FREQL*1.E9

200   KEY=2
      EFC=CMPLX(EF,-EF*TAND)
      IF(ABS(F-FM) .LT. .001E9) F=1.001*FM
      AK=-FM/F
      U=1.
      UEFF=(U*U-AK*AK)/U
          IF(F .LT. .999*FM)THEN
              UEFF=-UEFF
              KEY=1
              EFC=CONJG(EFC)
          ENDIF
      ZEFF=SQRT(U0*UEFF/E0/EFC)

      SIC1=PI/SQRT(3.)/1.84
      SIC=SIC1*ZD*ABS(AK/U)/ZEFF

      ZIBCWU=2.*ZD/(1.+(SIC/SI)*(SIC/SI))
      ZIBC=ZIBCWU*Z0/ZD

      SR=2.*PI*F*SQRT(UEFF*EFC)*R/C

      B0=BF(0,SR,KEY)
      A0=-BF(1,SR,KEY)

      IF(KEY.EQ.1)A0=-A0
```

```
      C1A=SI*B0/2./A0
      C1C=PI*ZD/2./ZEFF

      AL11=CMPLX(0.,1.)
      C11=C1A+AL11*C1C
      IF(F.LT.FM) C11=-C1A+AL11*C1C

      C1B=0.
      C2BR=0.
      C2BI=0.
      DO 100 N=1,NGFT
            RN=N
            BN=BF(N,SR,KEY)
            AN=(SR*BF(N-1,SR,KEY)-RN*BF(N,SR,KEY))/SR
            CK1=(SIN(RN*SI)*SIN(RN*SI))/(RN*RN*SI)
            CK2=RN*AK/U/SR
            CK22=CK2*CK2
            C1BD=AN*AN-CK22*BN*BN
            C1B=C1B+CK1*AN*BN/C1BD
            CK3=COS(2.*RN*PI/3.)
            CK4=SIN(2.*RN*PI/3.)
            C2BR=C2BR+CK1*(AN*BN*CK3)/C1BD
            C2BI=C2BI+CK1*CK2*BN*BN*CK4/C1BD
100   CONTINUE

      IF(F.LT.FM) C1B=-C1B
      C1=C11+C1B
      C2=C1A+C2BR-AL11*C2BI
      C3=C1A+C2BR+AL11*C2BI
      IF(F.LT.FM)THEN
        C2=-C2
        C3=-C3
      ENDIF
      C4=C1*C1-C2*C3
      C5=C1*C1*C1+C2*C2*C2+C3*C3*C3-3.*C1*C2*C3
      C6=C2*C2-C1*C3
      C7=C3*C3-C1*C2
      SM=PI*ZD/ZEFF/C5
      AL1=CMPLX(0.,-1.)
      AL2=SM*C4
      AL=1.+AL1*AL2
      BE=AL1*SM*C6
      GA=AL1*SM*C7

      ZINWU=-ZD+AL1*2.*ZEFF*C5/PI/C4
      ZIN=ZINWU*Z0/ZD

      RIN=REAL(ZIN)
      XIN=AIMAG(ZIN)

C     SALAY/PEPPIATT REACTANCE SIGN CORRECTION
      XIN=-XIN

      DBRTNL=20.*LOG10(CABS(AL))
      DBISOL=20.*LOG10(CABS(BE))
      DBINSL=20.*LOG10(CABS(GA))

      YIN=1./ZIN
      GIN=REAL(YIN)
      BIN=AIMAG(YIN)

C     SALAY/PEPPIATT ADMITTANCE SIGN CORRECTION
      BIN=-BIN
```

```
      WRITE(*,10)F/1.E9,RIN,XIN,DBRTNL,DBISOL,DBINSL,GIN,BIN
10    FORMAT(1X,6(1X,F8.2),2(1X,F8.4))

           F=F+FREQS*1.E9
           IF(F.GT.FREQH*1.00001E9)GO TO 210
      GO TO 200
210         STOP
      END

      COMPLEX FUNCTION BF(IP,Z,KEY)
      IF(IP.GT.20)GOTO 1450
      IF(Z.GT.20.)GOTO 1450
      X9=Z/2.
      Y9=-X9*X9
      IF(KEY.EQ.1) Y9=X9*X9
      J9=1
      RJ9=J9
      IF9=50
      IF(IP.EQ.0)GOTO 1340
      DO 1000 K9=1,IP
            RK9=K9
            RJ9=RJ9*X9/RK9
1000    CONTINUE
1340    T9=RJ9
      DO 2000 K9=1,IF9
            RK9=K9
            RIP=IP
            RJ9=RJ9*Y9/RK9/(RIP+RK9)
            S9=T9+RJ9
            IF(S9.EQ.T9)GOTO 1470
            T9=S9
2000    CONTINUE
      WRITE(*,*)'MAX NO BESSEL CALC ITERATIONS EXCEEDED.'
      WRITE(*,*)'SELECT NEW F,EF,4PIMS,OR R'
      STOP
1450  WRITE(*,*)'THE VALUE OF BESSEL ORDER OR ARGUMENT GT 20.'
      WRITE(*,*)'SELECT NEW F,EF,4PIMS,OR R.'
      STOP
1470  BF=T9
      RETURN
      END
```

Output from CIRCREN

```
FOR STRIPLINE ENTER 1 , FOR MICROSTRIP ENTER 2 :
1
ENTER NUMBER OF GREENS FUNCTION TERMS TO USE :
9
ENTER 4PIMS(G),DIEL CON FERR,LOSS TAN FERR :
1780,14,0
ENTER DIEL CON OUTSIDE FERR :
10
ENTER DIMS RADIUS,LINE W,DIEL TH,LINE TH (ALL CM):
.3,.417,.125,0
Z0=    7.862183
ENTER (IN GHZ): FLOW,FHIGH,FSTEP
2,20,.5
```

FREQ	RIN	XIN	RTN LS	ISOL	INSLS	GIN	BIN
2.00	6.57	0.90	-19.27	-18.26	-0.12	0.1493	-0.0205
2.50	6.79	1.09	-19.68	-18.72	-0.11	0.1436	-0.0229
3.00	7.06	1.23	-20.17	-19.29	-0.09	0.1374	-0.0240
3.50	7.39	1.33	-20.73	-19.96	-0.08	0.1311	-0.0235
4.00	7.76	1.35	-21.29	-20.70	-0.07	0.1250	-0.0217
4.50	8.17	1.27	-21.77	-21.42	-0.06	0.1195	-0.0186
5.00	8.59	1.09	-22.02	-22.01	-0.05	0.1146	-0.0145
5.50	8.98	0.77	-21.89	-22.26	-0.05	0.1105	-0.0094
6.00	9.30	0.32	-21.34	-22.00	-0.06	0.1074	-0.0037
6.50	9.49	-0.24	-20.44	-21.23	-0.07	0.1053	0.0027
7.00	9.52	-0.87	-19.34	-20.14	-0.09	0.1041	0.0096
7.50	9.37	-1.51	-18.17	-18.90	-0.12	0.1041	0.0168
8.00	9.02	-2.10	-17.01	-17.65	-0.16	0.1051	0.0245
8.50	8.52	-2.58	-15.88	-16.47	-0.22	0.1075	0.0326
9.00	7.91	-2.93	-14.78	-15.37	-0.28	0.1112	0.0411
9.50	7.22	-3.12	-13.68	-14.33	-0.36	0.1166	0.0505
10.00	6.48	-3.24	-12.41	-13.31	-0.48	0.1235	0.0617
10.50	5.41	-3.64	-9.92	-11.97	-0.79	0.1272	0.0857
11.00	4.93	0.20	-12.79	-12.06	-0.53	0.2024	-0.0081
11.50	5.10	-0.86	-13.03	-11.22	-0.58	0.1907	0.0323
12.00	4.75	-0.43	-12.07	-10.17	-0.75	0.2090	0.0188
12.50	4.55	0.36	-11.42	-9.01	-0.96	0.2184	-0.0175
13.00	4.58	1.45	-10.86	-7.64	-1.27	0.1983	-0.0629
13.50	5.03	2.94	-10.21	-6.04	-1.83	0.1483	-0.0866
14.00	6.44	5.00	-9.30	-4.34	-2.89	0.0969	-0.0752
14.50	10.65	7.29	-8.13	-2.85	-4.86	0.0640	-0.0438
15.00	19.91	3.26	-7.01	-1.95	-7.87	0.0489	-0.0080
15.50	16.57	-9.45	-6.19	-1.79	-10.13	0.0455	0.0260
16.00	8.50	-10.19	-5.52	-2.17	-9.50	0.0483	0.0579
16.50	1.28	-7.76	-1.43	-7.10	-10.71	0.0207	0.1255
17.00	4.26	-5.79	-5.90	-2.76	-6.70	0.0826	0.1120
17.50	3.18	-4.60	-5.22	-3.40	-6.16	0.1018	0.1471
18.00	2.50	-3.47	-4.66	-3.98	-5.88	0.1365	0.1897
18.50	1.95	-2.40	-3.99	-4.65	-5.89	0.2036	0.2509
19.00	1.30	-1.25	-2.83	-5.88	-6.58	0.4000	0.3841
19.50	0.04	1.30	-0.09	-19.96	-20.19	0.0238	-0.7708
20.00	4.47	2.75	-9.22	-3.66	-3.47	0.1623	-0.1000

STOP